Rules for integrands of the form $(a+b x)^m (c+d x)^n (e+f x)^p (g+h x)^q$
when $b c-a d\neq 0 \land b e-a f\neq 0 \land b g-a h\neq 0 \land d e-c f\neq 0 \land d g-c h\neq 0 \land f g-e h\neq 0$

1. $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x) \, dx$

   1. $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$

      1: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m\in \mathbb{Z}^+ \lor (m \mid n)\in \mathbb{Z}$

- **Derivation: Algebraic expansion**
- **Rule 1.1.1.4.1.1.1: If $m\in \mathbb{Z}^+ \lor (m \mid n)\in \mathbb{Z}$, then**

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$
$$\int \text{ExpandIntegrand}\left[(a+b x)^m (c+d x)^n (e+f x) (g+h x), x\right] dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n*(e+f*x)*(g+h*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && (IGtQ[m,0] || IntegersQ[m,n])
```

2: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m+n+2=0 \wedge m\neq -1$

- **Derivation: ???**
- **Rule 1.1.1.4.1.1.2: If $m+n+2=0 \wedge m\neq -1$, then**

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$

$$\frac{\left(b^2 d e g-a^2 d f h m-a b (d (f g+e h)-c f h (m+1))+b f h (b c-a d) (m+1) x\right) (a+b x)^{m+1} (c+d x)^{n+1}}{b^2 d (b c-a d) (m+1)}+$$

$$\frac{a d f h m+b (d (f g+e h)-c f h (m+2))}{b^2 d}\int (a+b x)^{m+1} (c+d x)^n \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  (b^2*d*e*g-a^2*d*f*h*m-a*b*(d*(f*g+e*h)-c*f*h*(m+1))+b*f*h*(b*c-a*d)*(m+1)*x)*(a+b*x)^(m+1)*(c+d*x)^(n+1)/
    (b^2*d*(b*c-a*d)*(m+1)) +
  (a*d*f*h*m+b*(d*(f*g+e*h)-c*f*h*(m+2)))/(b^2*d)*Int[(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[m+n+2,0] && NeQ[m,-1] && (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

3. $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m<-1$

1: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m<-1 \wedge n<-1$

Derivation: ???

Rule 1.1.1.4.1.1.3.1: If $m<-1 \wedge n<-1$, then

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$

$$\Big(b^2 c d e g (n+1)+a^2 c d f h (n+1)+a b \left(d^2 e g (m+1)+c^2 f h (m+1)-c d (f g+e h) (m+n+2)\right)+$$
$$\left(a^2 d^2 f h (n+1)-a b d^2 (f g+e h) (n+1)+b^2 \left(c^2 f h (m+1)-c d (f g+e h) (m+1)+d^2 e g (m+n+2)\right)\right) x\Big) \Big/ \left(b d (b c-a d)^2 (m+1) (n+1)\right)$$
$$(a+b x)^{m+1} (c+d x)^{n+1}-$$

$$\Big(a^2 d^2 f h \left(2+3 n+n^2\right)+a b d (n+1) (2 c f h (m+1)-d (f g+e h) (m+n+3))+$$
$$b^2 \left(c^2 f h \left(2+3 m+m^2\right)-c d (f g+e h) (m+1) (m+n+3)+d^2 e g \left(6+m^2+5 n+n^2+m (2 n+5)\right)\right)\Big) \Big/ \left(b d (b c-a d)^2 (m+1) (n+1)\right) \cdot$$

$$\int (a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,dx$$

– **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  (b^2*c*d*e*g*(n+1)+a^2*c*d*f*h*(n+1)+a*b*(d^2*e*g*(m+1)+c^2*f*h*(m+1)-c*d*(f*g+e*h)*(m+n+2))+
      (a^2*d^2*f*h*(n+1)-a*b*d^2*(f*g+e*h)*(n+1)+b^2*(c^2*f*h*(m+1)-c*d*(f*g+e*h)*(m+1)+d^2*e*g*(m+n+2)))*x)/
    (b*d*(b*c-a*d)^2*(m+1)*(n+1))*(a+b*x)^(m+1)*(c+d*x)^(n+1) -
  (a^2*d^2*f*h*(2+3*n+n^2)+a*b*d*(n+1)*(2*c*f*h*(m+1)-d*(f*g+e*h)*(m+n+3))+
      b^2*(c^2*f*h*(2+3*m+m^2)-c*d*(f*g+e*h)*(m+1)*(m+n+3)+d^2*e*g*(6+m^2+5*n+n^2+m*(2*n+5))))/
    (b*d*(b*c-a*d)^2*(m+1)*(n+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && LtQ[m,-1] && LtQ[n,-1]
```

2. $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m<-1 \wedge n \nless -1$

1: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m<-2$

Derivation: ???

Rule 1.1.1.4.1.1.3.2.1: If $m<-2$, then

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$

$$\Big(b^3 c e g (m+2) - a^3 d f h (n+2) - a^2 b (c f h m - d (f g + e h) (m+n+3)) - a b^2 (c (f g + e h) + d e g (2 m + n + 4)) + b \left(a^2 d f h (m-n) - a b (2 c f h (m+1) - d (f g + e h) (n+1)) + b^2 (c (f g + e h) (m+1) - d e g (m+n+2))\right) x\Big) \Big/ \left(b^2 (b c - a d)^2 (m+1) (m+2)\right) (a+b x)^{m+1} (c+d x)^{n+1} +$$

$$\left(\frac{f h}{b^2} - \left(d (m+n+3) \left(a^2 d f h (m-n) - a b (2 c f h (m+1) - d (f g + e h) (n+1)) + b^2 (c (f g + e h) (m+1) - d e g (m+n+2))\right)\right) \Big/ \left(b^2 (b c - a d)^2 (m+1) (m+2)\right)\right) \int (a+b x)^{m+2} (c+d x)^n \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  (b^3*c*e*g*(m+2)-a^3*d*f*h*(n+2)-a^2*b*(c*f*h*m-d*(f*g+e*h)*(m+n+3))-a*b^2*(c*(f*g+e*h)+d*e*g*(2*m+n+4))+
      b*(a^2*d*f*h*(m-n)-a*b*(2*c*f*h*(m+1)-d*(f*g+e*h)*(n+1))+b^2*(c*(f*g+e*h)*(m+1)-d*e*g*(m+n+2)))*x)/
    (b^2*(b*c-a*d)^2*(m+1)*(m+2))*(a+b*x)^(m+1)*(c+d*x)^(n+1) +
  (f*h/b^2-(d*(m+n+3)*(a^2*d*f*h*(m-n)-a*b*(2*c*f*h*(m+1)-d*(f*g+e*h)*(n+1))+b^2*(c*(f*g+e*h)*(m+1)-d*e*g*(m+n+2))))/
      (b^2*(b*c-a*d)^2*(m+1)*(m+2)))*
    Int[(a+b*x)^(m+2)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && (LtQ[m,-2] || EqQ[m+n+3,0] && Not[LtQ[n,-2]])
```

2: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $-2 \le m < -1$

Derivation: ???

Rule 1.1.1.4.1.1.3.2.2: If $-2 \le m < -1$, then

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$

$$\left(a^2 d f h (n+2) + b^2 d e g (m+n+3) + a b (c f h (m+1) - d (f g + e h) (m+n+3)) + b f h (b c - a d) (m+1) x\right) \Big/ \left(b^2 d (b c - a d) (m+1) (m+n+3)\right)$$

$$(a+b x)^{m+1} (c+d x)^{n+1} -$$

$$\left(a^2 d^2 f h (n+1) (n+2) + a b d (n+1) (2 c f h (m+1) - d (f g + e h) (m+n+3)) + b^2 \left(c^2 f h (m+1) (m+2) - c d (f g + e h) (m+1) (m+n+3) + d^2 e g (m+n+2) (m+n+3)\right)\right) \Big/ \left(b^2 d (b c - a d) (m+1) (m+n+3)\right) \int (a+b x)^{m+1} (c+d x)^n \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_.*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  (a^2*d*f*h*(n+2)+b^2*d*e*g*(m+n+3)+a*b*(c*f*h*(m+1)-d*(f*g+e*h)*(m+n+3))+b*f*h*(b*c-a*d)*(m+1)*x)/
    (b^2*d*(b*c-a*d)*(m+1)*(m+n+3))*(a+b*x)^(m+1)*(c+d*x)^(n+1) -
  (a^2*d^2*f*h*(n+1)*(n+2)+a*b*d*(n+1)*(2*c*f*h*(m+1)-d*(f*g+e*h)*(m+n+3))+
      b^2*(c^2*f*h*(m+1)*(m+2)-c*d*(f*g+e*h)*(m+1)*(m+n+3)+d^2*e*g*(m+n+2)*(m+n+3)))/
    (b^2*d*(b*c-a*d)*(m+1)*(m+n+3))*Int[(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && (GeQ[m,-2] && LtQ[m,-1] || SumSimplerQ[m,1]) && NeQ[m,-1] && NeQ[m+n+3,0]
```

4: $\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx$ when $m \nless -1 \land m+n+2 \neq 0 \land m+n+3 \neq 0$

Derivation: ???

Rule 1.1.1.4.1.1.4: If $m \nless -1 \land m+n+2 \neq 0 \land m+n+3 \neq 0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x) (g+h x) \, dx \longrightarrow$$

$$-\left((a d f h (n+2) + b c f h (m+2) - b d (f g + e h) (m+n+3) - b d f h (m+n+2) x) (a+b x)^{m+1} (c+d x)^{n+1}\right) \Big/ \left(b^2 d^2 (m+n+2) (m+n+3)\right) + \left(a^2 d^2 f h (n+1) (n+2) + a b d (n+1) (2 c f h (m+1) - d (f g + e h) (m+n+3)) + b^2 \left(c^2 f h (m+1) (m+2) - c d (f g + e h) (m+1) (m+n+3) + d^2 e g (m+n+2) (m+n+3)\right)\right) \Big/ \left(b^2 d^2 (m+n+2) (m+n+3)\right) \cdot \int (a+b x)^m (c+d x)^n \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_+f_.*x_)*(g_.+h_.*x_),x_Symbol] :=
  -(a*d*f*h*(n+2)+b*c*f*h*(m+2)-b*d*(f*g+e*h)*(m+n+3)-b*d*f*h*(m+n+2)*x)*(a+b*x)^(m+1)*(c+d*x)^(n+1)/
    (b^2*d^2*(m+n+2)*(m+n+3)) +
  (a^2*d^2*f*h*(n+1)*(n+2)+a*b*d*(n+1)*(2*c*f*h*(m+1)-d*(f*g+e*h)*(m+n+3))+
      b^2*(c^2*f*h*(m+1)*(m+2)-c*d*(f*g+e*h)*(m+1)*(m+n+3)+d^2*e*g*(m+n+2)*(m+n+3)))/
    (b^2*d^2*(m+n+2)*(m+n+3))*Int[(a+b*x)^m*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && NeQ[m+n+2,0] && NeQ[m+n+3,0]
```

2: $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\,dx$ when $(m \mid n \mid p) \in \mathbb{Z} \lor (n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.1.4.1.2: If $(m \mid n \mid p) \in \mathbb{Z} \lor (n \mid p) \in \mathbb{Z}^+$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\,dx \longrightarrow \int \text{ExpandIntegrand}[(a+b x)^m (c+d x)^n (e+f x)^p (g+h x),\,x]\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && (IntegersQ[m,n,p] || IGtQ[n,0] && IGtQ[p,0])
```

3. $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\,dx$ when $m < -1$

1: $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\,dx$ when $m < -1 \land n > 0$

Derivation: Nondegenerate trilinear recurrence 1

Rule 1.1.1.4.1.3.1: If $m < -1 \land n > 0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\,dx \longrightarrow$$

$$\frac{(b g-a h)\,(a+b x)^{m+1}\,(c+d x)^n\,(e+f x)^{p+1}}{b\,(b e-a f)\,(m+1)} - \frac{1}{b\,(b e-a f)\,(m+1)} \int (a+b x)^{m+1}\,(c+d x)^{n-1}\,(e+f x)^p \cdot$$
$$(b c\,(f g-e h)\,(m+1) + (b g-a h)\,(d e n+c f\,(p+1)) + d\,(b\,(f g-e h)\,(m+1) + f\,(b g-a h)\,(n+p+1))\,x)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  (b*g-a*h)*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^(p+1)/(b*(b*e-a*f)*(m+1)) -
  1/(b*(b*e-a*f)*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^p*
    Simp[b*c*(f*g-e*h)*(m+1)+(b*g-a*h)*(d*e*n+c*f*(p+1))+d*(b*(f*g-e*h)*(m+1)+f*(b*g-a*h)*(n+p+1))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,p},x] && ILtQ[m,-1] && GtQ[n,0]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  (b*g-a*h)*(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^(p+1)/(b*(b*e-a*f)*(m+1)) -
  1/(b*(b*e-a*f)*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1)*(e+f*x)^p*
    Simp[b*c*(f*g-e*h)*(m+1)+(b*g-a*h)*(d*e*n+c*f*(p+1))+d*(b*(f*g-e*h)*(m+1)+f*(b*g-a*h)*(n+p+1))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,p},x] && LtQ[m,-1] && GtQ[n,0] && IntegersQ[2*m,2*n,2*p]
```

2: $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x) \, dx$ when $m < -1 \land n \ngtr 0$

Derivation: Nondegenerate trilinear recurrence 3

Rule 1.1.1.4.1.3.2: If $m < -1 \land n \ngtr 0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x) \, dx \longrightarrow$$
$$\frac{(b g-a h) (a+b x)^{m+1} (c+d x)^{n+1} (e+f x)^{p+1}}{(m+1) (b c-a d) (b e-a f)} +$$
$$\frac{1}{(m+1) (b c-a d) (b e-a f)} \int (a+b x)^{m+1} (c+d x)^n (e+f x)^p \cdot$$
$$((a d f g-b (d e+c f) g+b c e h) (m+1)-(b g-a h) (d e (n+1)+c f (p+1))-d f (b g-a h) (m+n+p+3) x) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  (b*g-a*h)*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[(a*d*f*g-b*(d*e+c*f)*g+b*c*e*h)*(m+1)-(b*g-a*h)*(d*e*(n+1)+c*f*(p+1))-d*f*(b*g-a*h)*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x] && ILtQ[m,-1]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  (b*g-a*h)*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[(a*d*f*g-b*(d*e+c*f)*g+b*c*e*h)*(m+1)-(b*g-a*h)*(d*e*(n+1)+c*f*(p+1))-d*f*(b*g-a*h)*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x] && LtQ[m,-1] && IntegersQ[2*m,2*n,2*p]
```

4: $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x) \, dx$ when $m > 0 \land m+n+p+2 \neq 0$

Derivation: Nondegenerate trilinear recurrence 2

Rule 1.1.1.4.1.4: If $m > 0 \land m+n+p+2 \neq 0$, then

$$\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)\, dx \longrightarrow$$
$$\frac{h (a+b x)^m (c+d x)^{n+1} (e+f x)^{p+1}}{d f (m+n+p+2)}+$$
$$\frac{1}{d f (m+n+p+2)} \int (a+b x)^{m-1} (c+d x)^n (e+f x)^p \cdot$$
$$(a d f g (m+n+p+2)-h (b c e m+a (d e (n+1)+c f (p+1)))+(b d f g (m+n+p+2)+h (a d f m-b (d e (m+n+1)+c f (m+p+1)))) x)\, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  h*(a+b*x)^m*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(m+n+p+2)) +
  1/(d*f*(m+n+p+2))*Int[(a+b*x)^(m-1)*(c+d*x)^n*(e+f*x)^p*
    Simp[a*d*f*g*(m+n+p+2)-h*(b*c*e*m+a*(d*e*(n+1)+c*f*(p+1)))+(b*d*f*g*(m+n+p+2)+h*(a*d*f*m-b*(d*e*(m+n+1)+c*f*(m+p+1))))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x] && GtQ[m,0] && NeQ[m+n+p+2,0] && IntegerQ[m]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  h*(a+b*x)^m*(c+d*x)^(n+1)*(e+f*x)^(p+1)/(d*f*(m+n+p+2)) +
  1/(d*f*(m+n+p+2))*Int[(a+b*x)^(m-1)*(c+d*x)^n*(e+f*x)^p*
    Simp[a*d*f*g*(m+n+p+2)-h*(b*c*e*m+a*(d*e*(n+1)+c*f*(p+1)))+(b*d*f*g*(m+n+p+2)+h*(a*d*f*m-b*(d*e*(m+n+1)+c*f*(m+p+1))))*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x] && GtQ[m,0] && NeQ[m+n+p+2,0] && IntegersQ[2*m,2*n,2*p]
```

5: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)\,dx$ when $m+n+p+2\in\mathbb{Z}^-\wedge m\neq -1$

Derivation: Nondegenerate trilinear recurrence 3

Note: If $m+n+p+2\in\mathbb{Z}^-$**, then** $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)\,dx$ **can be expressed in terms of the hypergeometric function 2F1.**

Rule 1.1.1.4.1.5: If $m+n+p+2\in\mathbb{Z}^-\wedge m\neq -1$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)\,dx\;\longrightarrow$$

$$\frac{(b\,g-a\,h)\,(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}\,(e+f\,x)^{p+1}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}+$$

$$\frac{1}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\cdot$$

$$((a\,d\,f\,g-b\,(d\,e+c\,f)\,g+b\,c\,e\,h)\,(m+1)-(b\,g-a\,h)\,(d\,e\,(n+1)+c\,f\,(p+1))-d\,f\,(b\,g-a\,h)\,(m+n+p+3)\,x)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  (b*g-a*h)*(a+b*x)^(m+1)*(c+d*x)^(n+1)*(e+f*x)^(p+1)/((m+1)*(b*c-a*d)*(b*e-a*f)) +
  1/((m+1)*(b*c-a*d)*(b*e-a*f))*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*
    Simp[(a*d*f*g-b*(d*e+c*f)*g+b*c*e*h)*(m+1)-(b*g-a*h)*(d*e*(n+1)+c*f*(p+1))-d*f*(b*g-a*h)*(m+n+p+3)*x,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x] && ILtQ[m+n+p+2,0] && NeQ[m,-1] &&
  (SumSimplerQ[m,1] || Not[NeQ[n,-1] && SumSimplerQ[n,1]] && Not[NeQ[p,-1] && SumSimplerQ[p,1]])
```

6. $\int \frac{(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)}{a+b\,x}\,dx$

?: $\int \frac{(a+b\,x)^m\,(c+d\,x)^n\,(g+h\,x)}{e+f\,x}\,dx$ when $m+n+1\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $\frac{(a+b\,x)^m\,(c+d\,x)^n\,(g+h\,x)}{e+f\,x}==\frac{(f\,g-e\,h)\,(c\,f-d\,e)^{m+n+1}\,(a+b\,x)^m}{f^{m+n+2}\,(c+d\,x)^{m+1}\,(e+f\,x)}+\frac{(a+b\,x)^m}{f^{m+n+2}\,(c+d\,x)^{m+1}}\,\frac{f^{m+n+2}\,(c+d\,x)^{m+n+1}\,(g+h\,x)-(f\,g-e\,h)\,(c\,f-d\,e)^{m+n+1}}{e+f\,x}$
- **Note: If** $m+n+1\in\mathbb{Z}^+$**, then** $\frac{f^{m+n+2}\,(c+d\,x)^{m+n+1}\,(g+h\,x)-(f\,g-e\,h)\,(c\,f-d\,e)^{m+n+1}}{e+f\,x}$ **is a polynomial in x.**

Rule 1.1.1.3.9.3: If $m+n+1\in\mathbb{Z}^+$**, then**

$$\int \frac{(a+b\,x)^m\,(c+d\,x)^n\,(g+h\,x)}{e+f\,x}\,dx\;\longrightarrow$$

$$\frac{(f g-e h)\,(c f-d e)^{m+n+1}}{f^{m+n+2}}\int\frac{(a+b x)^m}{(c+d x)^{m+1}\,(e+f x)}\,dx+\frac{1}{f^{m+n+2}}\int\frac{(a+b x)^m}{(c+d x)^{m+1}}\,\frac{f^{m+n+2}\,(c+d x)^{m+n+1}\,(g+h x)-(f g-e h)\,(c f-d e)^{m+n+1}}{e+f x}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(g_.+h_.*x_)/(e_.+f_.*x_),x_Symbol] :=
  (f*g-e*h)*(c*f-d*e)^(m+n+1)/f^(m+n+2)*Int[(a+b*x)^m/((c+d*x)^(m+1)*(e+f*x)),x] +
  1/f^(m+n+2)*Int[(a+b*x)^m/(c+d*x)^(m+1)*
    ExpandToSum[(f^(m+n+2)*(c+d*x)^(m+n+1)*(g+h*x)-(f*g-e*h)*(c*f-d*e)^(m+n+1))/(e+f*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IGtQ[m+n+1,0] && (LtQ[m,0] || SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

1: $\int\frac{(e+f x)^p\,(g+h x)}{(a+b x)\,(c+d x)}\,dx$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{g+h x}{(a+b x)\,(c+d x)}==\frac{b g-a h}{(b c-a d)\,(a+b x)}-\frac{d g-c h}{(b c-a d)\,(c+d x)}$

- **Rule 1.1.1.4.1.6.1:**

$$\int\frac{(e+f x)^p\,(g+h x)}{(a+b x)\,(c+d x)}\,dx\longrightarrow\frac{b g-a h}{b c-a d}\int\frac{(e+f x)^p}{a+b x}\,dx-\frac{d g-c h}{b c-a d}\int\frac{(e+f x)^p}{c+d x}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_*(g_.+h_.*x_)/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  (b*g-a*h)/(b*c-a*d)*Int[(e+f*x)^p/(a+b*x),x] -
  (d*g-c*h)/(b*c-a*d)*Int[(e+f*x)^p/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

2: $\int \frac{(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)}{a+b\,x}\,dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{g+h\,x}{a+b\,x} == \frac{h}{b} + \frac{b\,g-a\,h}{b\,(a+b\,x)}$

Rule 1.1.1.4.1.6.2:

$$\int \frac{(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)}{a+b\,x}\,dx \longrightarrow \frac{h}{b}\int (c+d\,x)^n\,(e+f\,x)^p\,dx + \frac{b\,g-a\,h}{b}\int \frac{(c+d\,x)^n\,(e+f\,x)^p}{a+b\,x}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_)/(a_.+b_.*x_),x_Symbol] :=
  h/b*Int[(c+d*x)^n*(e+f*x)^p,x] + (b*g-a*h)/b*Int[(c+d*x)^n*(e+f*x)^p/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,n,p},x]
```

7: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)\,dx$

Derivation: Algebraic expansion

- **Basis:** $g+h\,x == \frac{h\,(a+b\,x)}{b} + \frac{b\,g-a\,h}{b}$
- **Note:** For $\frac{g+h\,x}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}}$, ensuring the simpler square-root factors remain in the denominator of the resulting integrands causes the two elliptic integrals in the antiderivative to have the same and simplest arguments.

Rule 1.1.1.4.1.7:

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)\,dx \longrightarrow$$
$$\frac{h}{b}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,dx + \frac{b\,g-a\,h}{b}\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,dx$$

Program code:

```
Int[(g_.+h_.*x_)/(Sqrt[a_.+b_.*x_]*Sqrt[c_+d_.*x_]*Sqrt[e_+f_.*x_]),x_Symbol] :=
  h/f*Int[Sqrt[e+f*x]/(Sqrt[a+b*x]*Sqrt[c+d*x]),x] + (f*g-e*h)/f*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && SimplerQ[a+b*x,e+f*x] && SimplerQ[c+d*x,e+f*x]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_),x_Symbol] :=
  h/b*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p,x] + (b*g-a*h)/b*Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]] && Not[SumSimplerQ[p,1]])
```

2. $\int (a+b x)^m (c+d x)^n (e+f x)^p (g+h x)^q \, dx$ when $2 m \in \mathbb{Z} \bigwedge n^2 == \frac{1}{4} \bigwedge p^2 == \frac{1}{4} \bigwedge q^2 == \frac{1}{4}$

1. $\int (a+b x)^m (c+d x)^n \sqrt{e+f x} \sqrt{g+h x} \, dx$ when $2 m \in \mathbb{Z} \bigwedge n^2 == \frac{1}{4}$

1. $\int (a+b x)^m \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x} \, dx$ when $2 m \in \mathbb{Z}$

1: $\int (a+b x)^m \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x} \, dx$ when $2 m \in \mathbb{Z} \wedge m < -1$

Derivation: Integration by parts

- **Basis:** $\partial_x \left(\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}\right) == \frac{d e g+c f g+c e h+2 (d f g+d e h+c f h) x+3 d f h x^2}{2 \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}$

- **Rule 1.1.1.4.2.1.1.1: If $2 m \in \mathbb{Z} \wedge m < -1$, then**

$$\int (a+b x)^m \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x} \, dx \longrightarrow$$

$$\frac{(a+b x)^{m+1} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{b (m+1)} - \frac{1}{2 b (m+1)} \int \frac{(a+b x)^{m+1} \left(d e g+c f g+c e h+2 (d f g+d e h+c f h) x+3 d f h x^2\right)}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_],x_Symbol] :=
  (a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(b*(m+1)) -
  1/(2*b*(m+1))*Int[(a+b*x)^(m+1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
    Simp[d*e*g+c*f*g+c*e*h+2*(d*f*g+d*e*h+c*f*h)*x+3*d*f*h*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && LtQ[m,-1]
```

2: $\int (a+b x)^m \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x} \, dx$ when $2 m \in \mathbb{Z} \wedge m \nless -1$

Rule 1.1.1.4.2.1.1.2: If $2 m \in \mathbb{Z} \bigwedge m \nless -1$, then

$$\int (a+b x)^m \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x} \, dx \longrightarrow$$

$$\frac{2 (a+b x)^{m+1} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{b (2 m+5)}+\frac{1}{b (2 m+5)} \int \frac{(a+b x)^m}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \cdot \\ \left(3 b c e g-a (d e g+c f g+c e h)+2 (b (d e g+c f g+c e h)-a (d f g+d e h+c f h)) x-(3 a d f h-b (d f g+d e h+c f h)) x^2\right) dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_],x_Symbol] :=
  2*(a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(b*(2*m+5)) +
  1/(b*(2*m+5))*Int[((a+b*x)^m)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x])*
     Simp[3*b*c*e*g-a*(d*e*g+c*f*g+c*e*h)+2*(b*(d*e*g+c*f*g+c*e*h)-a*(d*f*g+d*e*h+c*f*h))*x-(3*a*d*f*h-b*(d*f*g+d*e*h+c*f*h))*x^2,x
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && Not[LtQ[m,-1]]
```

2. $\int \frac{(a+b\,x)^m \sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx$ when $2\,m \in \mathbb{Z}$

1: $\int \frac{(a+b\,x)^m \sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx$ when $2\,m \in \mathbb{Z} \wedge m > 0$

Rule 1.1.1.4.2.1.2.1: If $2\,m \in \mathbb{Z} \wedge m > 0$, then

$$\int \frac{(a+b\,x)^m \sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx \longrightarrow$$

$$\frac{2\,(a+b\,x)^m \sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{d\,(2\,m+3)} - \frac{1}{d\,(2\,m+3)} \int \frac{(a+b\,x)^{m-1}}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}} \cdot \\ \big(2\,b\,c\,e\,g\,m + a\,(c\,(f\,g+e\,h) - 2\,d\,e\,g\,(m+1)) - \\ (b\,(2\,d\,e\,g - c\,(f\,g+e\,h)\,(2\,m+1)) - a\,(2\,c\,f\,h - d\,(2\,m+1)\,(f\,g+e\,h)))\,x - \\ (2\,a\,d\,f\,h\,m + b\,(d\,(f\,g+e\,h) - 2\,c\,f\,h\,(m+1)))\,x^2\big)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  2*(a+b*x)^m*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(d*(2*m+3)) -
  1/(d*(2*m+3))*Int[((a+b*x)^(m-1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[2*b*c*e*g*m+a*(c*(f*g+e*h)-2*d*e*g*(m+1)) -
      (b*(2*d*e*g-c*(f*g+e*h)*(2*m+1))-a*(2*c*f*h-d*(2*m+1)*(f*g+e*h)))*x -
      (2*a*d*f*h*m+b*(d*(f*g+e*h)-2*c*f*h*(m+1)))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && GtQ[m,0]
```

2. $\int \frac{(a+b\,x)^m\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx$ when $2\,m\in\mathbb{Z}\,\wedge\,m<0$

1: $\int \frac{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{(a+b\,x)\,\sqrt{c+d\,x}}\,dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{a+b\,x}==\frac{(b\,e-a\,f)\,(b\,g-a\,h)}{b^2\,(a+b\,x)\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}+\frac{b\,f\,g+b\,e\,h-a\,f\,h+b\,f\,h\,x}{b^2\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}$

Rule 1.1.1.4.2.1.2.2.1:

$$\int \frac{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{(a+b\,x)\,\sqrt{c+d\,x}}\,dx\,\longrightarrow\,\frac{(b\,e-a\,f)\,(b\,g-a\,h)}{b^2}\int\frac{1}{(a+b\,x)\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx+\frac{1}{b^2}\int\frac{b\,f\,g+b\,e\,h-a\,f\,h+b\,f\,h\,x}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$$

Program code:

```
Int[Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  (b*e-a*f)*(b*g-a*h)/b^2*Int[1/((a+b*x)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  1/b^2*Int[Simp[b*f*g+b*e*h-a*f*h+b*f*h*x,x]/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

2: $\int \frac{(a+b\,x)^m\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx$ when $2\,m\in\mathbb{Z}\,\wedge\,m<-1$

Rule 1.1.1.4.2.1.2.2.2: If $2\,m\in\mathbb{Z}\,\wedge\,m<-1$, then

$$\int \frac{(a+b\,x)^m\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{\sqrt{c+d\,x}}\,dx\,\longrightarrow$$

$$\frac{(a+b\,x)^{m+1}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{(m+1)\,(b\,c-a\,d)}-\frac{1}{2\,(m+1)\,(b\,c-a\,d)}\int\frac{(a+b\,x)^{m+1}}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,\cdot$$
$$\left(c\,(f\,g+e\,h)+d\,e\,g\,(2\,m+3)+2\,(c\,f\,h+d\,(m+2)\,(f\,g+e\,h))\,x+d\,f\,h\,(2\,m+5)\,x^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  (a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/((m+1)*(b*c-a*d)) -
  1/(2*(m+1)*(b*c-a*d))*Int[((a+b*x)^(m+1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[c*(f*g+e*h)+d*e*g*(2*m+3)+2*(c*f*h+d*(m+2)*(f*g+e*h))*x+d*f*h*(2*m+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && LtQ[m,-1]
```

2. $\int \frac{(a+bx)^m (c+dx)^n}{\sqrt{e+fx}\,\sqrt{g+hx}}\,dx$ when $2m \in \mathbb{Z} \bigwedge n^2 == \frac{1}{4}$

1. $\int \frac{(a+bx)^m}{\sqrt{c+dx}\,\sqrt{e+fx}\,\sqrt{g+hx}}\,dx$ when $2m \in \mathbb{Z}$

1. $\int \frac{(a+bx)^m}{\sqrt{c+dx}\,\sqrt{e+fx}\,\sqrt{g+hx}}\,dx$ when $2m \in \mathbb{Z} \bigwedge m > 0$

1: $\int \frac{\sqrt{a+bx}}{\sqrt{c+dx}\,\sqrt{e+fx}\,\sqrt{g+hx}}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{(a+bx)\sqrt{\frac{(bg-ah)(c+dx)}{(dg-ch)(a+bx)}}\sqrt{\frac{(bg-ah)(e+fx)}{(fg-eh)(a+bx)}}}{\sqrt{c+dx}\,\sqrt{e+fx}} == 0$

- **Basis:** $\frac{1}{\sqrt{a+bx}\,\sqrt{\frac{(bg-ah)(c+dx)}{(dg-ch)(a+bx)}}\,\sqrt{\frac{(bg-ah)(e+fx)}{(fg-eh)(a+bx)}}\,\sqrt{g+hx}} == 2\,\text{Subst}\left[\frac{1}{(h-bx^2)\sqrt{1+\frac{(bc-ad)x^2}{dg-ch}}\sqrt{1+\frac{(be-af)x^2}{fg-eh}}},\,x,\,\frac{\sqrt{g+hx}}{\sqrt{a+bx}}\right]\partial_x\frac{\sqrt{g+hx}}{\sqrt{a+bx}}$

- **Rule 1.1.1.4.2.2.1.1.1:**

$$\int \frac{\sqrt{a+bx}}{\sqrt{c+dx}\,\sqrt{e+fx}\,\sqrt{g+hx}}\,dx \to \frac{(a+bx)\sqrt{\frac{(bg-ah)(c+dx)}{(dg-ch)(a+bx)}}\sqrt{\frac{(bg-ah)(e+fx)}{(fg-eh)(a+bx)}}}{\sqrt{c+dx}\,\sqrt{e+fx}}\int \frac{1}{\sqrt{a+bx}\,\sqrt{\frac{(bg-ah)(c+dx)}{(dg-ch)(a+bx)}}\,\sqrt{\frac{(bg-ah)(e+fx)}{(fg-eh)(a+bx)}}\,\sqrt{g+hx}}\,dx$$

$$\to \frac{2(a+bx)\sqrt{\frac{(bg-ah)(c+dx)}{(dg-ch)(a+bx)}}\sqrt{\frac{(bg-ah)(e+fx)}{(fg-eh)(a+bx)}}}{\sqrt{c+dx}\,\sqrt{e+fx}}\,\text{Subst}\left[\int\frac{1}{(h-bx^2)\sqrt{1+\frac{(bc-ad)x^2}{dg-ch}}\sqrt{1+\frac{(be-af)x^2}{fg-eh}}}\,dx,\,x,\,\frac{\sqrt{g+hx}}{\sqrt{a+bx}}\right]$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*x_]/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*(a+b*x)*Sqrt[(b*g-a*h)*(c+d*x)/((d*g-c*h)*(a+b*x))]*Sqrt[(b*g-a*h)*(e+f*x)/((f*g-e*h)*(a+b*x))]/(Sqrt[c+d*x]*Sqrt[e+f*x])*
    Subst[Int[1/((h-b*x^2)*Sqrt[1+(b*c-a*d)*x^2/(d*g-c*h)]*Sqrt[1+(b*e-a*f)*x^2/(f*g-e*h)]),x],x,Sqrt[g+h*x]/Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

2: $\int \frac{(a+b x)^{3/2}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{(a+b x)^{3/2}}{\sqrt{c+d x}} = \frac{b \sqrt{a+b x} \sqrt{c+d x}}{d} - \frac{(b c-a d) \sqrt{a+b x}}{d \sqrt{c+d x}}$

Rule 1.1.1.4.2.2.1.1.2:

$$\int \frac{(a+b x)^{3/2}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx \to \frac{b}{d} \int \frac{\sqrt{a+b x} \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx - \frac{(b c-a d)}{d} \int \frac{\sqrt{a+b x}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^(3/2)/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  b/d*Int[Sqrt[a+b*x]*Sqrt[c+d*x]/(Sqrt[e+f*x]*Sqrt[g+h*x]),x] -
  (b*c-a*d)/d*Int[Sqrt[a+b*x]/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

3: $\int \frac{(a+b x)^m}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$ when $2 m \in \mathbb{Z} \land m \geq 2$

Rule 1.1.1.4.2.2.1.1.3: If $2 m \in \mathbb{Z} \land m \geq 2$, then

$$\int \frac{(a+b x)^m}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx \to$$

$$\frac{2 b^2 (a+b x)^{m-2} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{d f h (2 m-1)} - \frac{1}{d f h (2 m-1)} \int \frac{(a+b x)^{m-3}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, .$$

$$\Big(a b^2 (d e g+c f g+c e h)+2 b^3 c e g (m-2)-a^3 d f h (2 m-1)+$$

$$b \left(2 a b (d f g+d e h+c f h)+b^2 (2 m-3) (d e g+c f g+c e h)-3 a^2 d f h (2 m-1)\right) x-$$

$$2\,b^2\,(m-1)\,(3\,a\,d\,f\,h-b\,(d\,f\,g+d\,e\,h+c\,f\,h))\,x^2\Big)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*b^2*(a+b*x)^(m-2)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(d*f*h*(2*m-1)) -
  1/(d*f*h*(2*m-1))*Int[((a+b*x)^(m-3)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[a*b^2*(d*e*g+c*f*g+c*e*h)+2*b^3*c*e*g*(m-2)-a^3*d*f*h*(2*m-1) +
      b*(2*a*b*(d*f*g+d*e*h+c*f*h)+b^2*(2*m-3)*(d*e*g+c*f*g+c*e*h)-3*a^2*d*f*h*(2*m-1))*x -
      2*b^2*(m-1)*(3*a*d*f*h-b*(d*f*g+d*e*h+c*f*h))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IntegerQ[2*m] && GeQ[m,2]
```

2. $\displaystyle\int \frac{(a+b\,x)^m}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m\in\mathbb{Z}\,\wedge\,m<0$

1: $\displaystyle\int \frac{1}{(a+b\,x)\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$

Derivation: Integration by substitution

- **Basis:** $\frac{F[x]}{\sqrt{c+d\,x}} = \frac{2}{d}\,\text{Subst}\left[F\left[-\frac{c-x^2}{d}\right],\,x,\,\sqrt{c+d\,x}\right]\,\partial_x\sqrt{c+d\,x}$

Rule 1.1.1.4.2.2.1.2.1:

$$\int \frac{1}{(a+b\,x)\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx \,\rightarrow\, -2\,\text{Subst}\left[\int \frac{1}{\left(b\,c-a\,d-b\,x^2\right)\,\sqrt{\frac{d\,e-c\,f}{d}+\frac{f\,x^2}{d}}\,\sqrt{\frac{d\,g-c\,h}{d}+\frac{h\,x^2}{d}}}\,dx,\,x,\,\sqrt{c+d\,x}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -2*Subst[Int[1/(Simp[b*c-a*d-b*x^2,x]*Sqrt[Simp[(d*e-c*f)/d+f*x^2/d,x]]*Sqrt[Simp[(d*g-c*h)/d+h*x^2/d,x]]),x],x,Sqrt[c+d*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && GtQ[(d*e-c*f)/d,0]
```

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -2*Subst[Int[1/(Simp[b*c-a*d-b*x^2,x]*Sqrt[Simp[(d*e-c*f)/d+f*x^2/d,x]]*Sqrt[Simp[(d*g-c*h)/d+h*x^2/d,x]]),x],x,Sqrt[c+d*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && Not[SimplerQ[e+f*x,c+d*x]] && Not[SimplerQ[g+h*x,c+d*x]]
```

x: $\int \frac{1}{\sqrt{a+b\,x}\ \sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{e+f\,x}\ \sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}}{\sqrt{c+d\,x}\ \sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}} == 0$

- **Basis:** $\frac{1}{(a+b\,x)^{3/2}\ \sqrt{g+h\,x}\ \sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}\ \sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}} == -\frac{2}{b\,g-a\,h}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{(b\,c-a\,d)\,x^2}{d\,g-c\,h}}\ \sqrt{1+\frac{(b\,e-a\,f)\,x^2}{f\,g-e\,h}}},\ x,\ \frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}\right]\,\partial_x \frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}$

Rule 1.1.1.4.2.2.1.2.2:

$$\int \frac{1}{\sqrt{a+b\,x}\ \sqrt{c+d\,x}\ \sqrt{e+f\,x}\ \sqrt{g+h\,x}}\,dx \longrightarrow \frac{(a+b\,x)\,\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}\ \sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}{\sqrt{c+d\,x}\ \sqrt{e+f\,x}} \int \frac{1}{(a+b\,x)^{3/2}\ \sqrt{g+h\,x}\ \sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}\ \sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}\,dx$$

$$\longrightarrow -\frac{2\,(a+b\,x)\,\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}\ \sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}{(b\,g-a\,h)\,\sqrt{c+d\,x}\ \sqrt{e+f\,x}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1+\frac{(b\,c-a\,d)\,x^2}{d\,g-c\,h}}\ \sqrt{1+\frac{(b\,e-a\,f)\,x^2}{f\,g-e\,h}}}\,dx,\ x,\ \frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}\right]$$

Program code:

```
(* Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -2*(a+b*x)*Sqrt[(b*g-a*h)*(c+d*x)/((d*g-c*h)*(a+b*x))]*Sqrt[(b*g-a*h)*(e+f*x)/((f*g-e*h)*(a+b*x))]/
    ((b*g-a*h)*Sqrt[c+d*x]*Sqrt[e+f*x])*
    Subst[Int[1/(Sqrt[1+(b*c-a*d)*x^2/(d*g-c*h)]*Sqrt[1+(b*e-a*f)*x^2/(f*g-e*h)]),x],x,Sqrt[g+h*x]/Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] *)
```

2: $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}\,\sqrt{g+h x}}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{g+h x}\,\sqrt{\frac{(b e-a f)\,(c+d x)}{(d e-c f)\,(a+b x)}}}{\sqrt{c+d x}\,\sqrt{-\frac{(b e-a f)\,(g+h x)}{(f g-e h)\,(a+b x)}}} = 0$

- **Basis:** $\frac{1}{(a+b x)^{3/2}\,\sqrt{e+f x}\,\sqrt{\frac{(b e-a f)\,(c+d x)}{(d e-c f)\,(a+b x)}}\,\sqrt{\frac{(-b e+a f)\,(g+h x)}{(f g-e h)\,(a+b x)}}} = -\frac{2}{b e-a f}\,\text{Subst}\left[\frac{1}{\sqrt{1+\frac{(b c-a d)\,x^2}{d e-c f}}\,\sqrt{1-\frac{(b g-a h)\,x^2}{f g-e h}}},\,x,\,\frac{\sqrt{e+f x}}{\sqrt{a+b x}}\right]\partial_x \frac{\sqrt{e+f x}}{\sqrt{a+b x}}$

Rule 1.1.1.4.2.2.1.2.2:

$$\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}\,\sqrt{e+f x}\,\sqrt{g+h x}}\,dx \to -\frac{(b e-a f)\,\sqrt{g+h x}\,\sqrt{\frac{(b e-a f)\,(c+d x)}{(d e-c f)\,(a+b x)}}}{(f g-e h)\,\sqrt{c+d x}\,\sqrt{-\frac{(b e-a f)\,(g+h x)}{(f g-e h)\,(a+b x)}}}\int \frac{1}{(a+b x)^{3/2}\,\sqrt{e+f x}\,\sqrt{\frac{(b e-a f)\,(c+d x)}{(d e-c f)\,(a+b x)}}\,\sqrt{\frac{(-b e+a f)\,(g+h x)}{(f g-e h)\,(a+b x)}}}\,dx$$

$$\to \frac{2\,\sqrt{g+h x}\,\sqrt{\frac{(b e-a f)\,(c+d x)}{(d e-c f)\,(a+b x)}}}{(f g-e h)\,\sqrt{c+d x}\,\sqrt{-\frac{(b e-a f)\,(g+h x)}{(f g-e h)\,(a+b x)}}}\,\text{Subst}\left[\int \frac{1}{\sqrt{1+\frac{(b c-a d)\,x^2}{d e-c f}}\,\sqrt{1-\frac{(b g-a h)\,x^2}{f g-e h}}}\,dx,\,x,\,\frac{\sqrt{e+f x}}{\sqrt{a+b x}}\right]$$

Program code:

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*Sqrt[g+h*x]*Sqrt[(b*e-a*f)*(c+d*x)/((d*e-c*f)*(a+b*x))]/
    ((f*g-e*h)*Sqrt[c+d*x]*Sqrt[-(b*e-a*f)*(g+h*x)/((f*g-e*h)*(a+b*x))])*
    Subst[Int[1/(Sqrt[1+(b*c-a*d)*x^2/(d*e-c*f)]*Sqrt[1-(b*g-a*h)*x^2/(f*g-e*h)]),x],x,Sqrt[e+f*x]/Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

3: $\int \frac{1}{(a+b x)^{3/2} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{(a+b x)^{3/2} \sqrt{c+d x}} = -\frac{d}{(b c-a d) \sqrt{a+b x} \sqrt{c+d x}} + \frac{b \sqrt{c+d x}}{(b c-a d) (a+b x)^{3/2}}$

Rule 1.1.1.4.2.2.1.2.3:

$$\int \frac{1}{(a+b x)^{3/2} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx \longrightarrow$$

$$-\frac{d}{b c-a d} \int \frac{1}{\sqrt{a+b x} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx + \frac{b}{b c-a d} \int \frac{\sqrt{c+d x}}{(a+b x)^{3/2} \sqrt{e+f x} \sqrt{g+h x}} \, dx$$

Program code:

```
Int[1/((a_.+b_.*x_)^(3/2)*Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -d/(b*c-a*d)*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  b/(b*c-a*d)*Int[Sqrt[c+d*x]/((a+b*x)^(3/2)*Sqrt[e+f*x]*Sqrt[g+h*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

4: $\int \frac{(a+b\,x)^m}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m\in\mathbb{Z}\,\wedge\,m\le -2$

Rule 1.1.1.4.2.2.1.2.4: If $2\,m\in\mathbb{Z}\,\wedge\,m\le -2$, then

$$\int \frac{(a+b\,x)^m}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx\,\longrightarrow$$

$$\frac{b^2\,(a+b\,x)^{m+1}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)\,(b\,g-a\,h)}-\frac{1}{2\,(m+1)\,(b\,c-a\,d)\,(b\,e-a\,f)\,(b\,g-a\,h)}\int\frac{(a+b\,x)^{m+1}}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\cdot$$
$$\big(2\,a^2\,d\,f\,h\,(m+1)-2\,a\,b\,(m+1)\,(d\,f\,g+d\,e\,h+c\,f\,h)+$$
$$b^2\,(2\,m+3)\,(d\,e\,g+c\,f\,g+c\,e\,h)-2\,b\,(a\,d\,f\,h\,(m+1)-b\,(m+2)\,(d\,f\,g+d\,e\,h+c\,f\,h))\,x+d\,f\,h\,b^2\,(2\,m+5)\,x^2\big)\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_/(Sqrt[c_.+d_.*x_]*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  b^2*(a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/((m+1)*(b*c-a*d)*(b*e-a*f)*(b*g-a*h)) -
  1/(2*(m+1)*(b*c-a*d)*(b*e-a*f)*(b*g-a*h))*Int[((a+b*x)^(m+1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[2*a^2*d*f*h*(m+1)-2*a*b*(m+1)*(d*f*g+d*e*h+c*f*h)+b^2*(2*m+3)*(d*e*g+c*f*g+c*e*h) -
      2*b*(a*d*f*h*(m+1)-b*(m+2)*(d*f*g+d*e*h+c*f*h))*x + d*f*h*(2*m+5)*b^2*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IntegerQ[2*m] && LeQ[m,-2]
```

2. $\int \frac{(a+b\,x)^m \sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m \in \mathbb{Z}$

1. $\int \frac{(a+b\,x)^m \sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $2\,m \in \mathbb{Z} \land m > 0$

1: $\int \frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}} == \partial_x \frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{g+h\,x}}{h\sqrt{e+f\,x}} + \frac{(d\,e-c\,f)\,(b\,f\,g+b\,e\,h-2\,a\,f\,h)}{2\,f^2\,h\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}} + \frac{(a\,d\,f\,h-b\,(d\,f\,g+d\,e\,h-c\,f\,h))\,\sqrt{e+f\,x}}{2\,f^2\,h\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{g+h\,x}} - \frac{(d\,e-c\,f)\,(f\,g-e\,h)\,\sqrt{a+b\,x}}{2\,f\,h\sqrt{c+d\,x}\,(e+f\,x)^{3/2}\,\sqrt{g+h\,x}}$
- **Basis:** $\frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}} == \partial_x \frac{b\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}{f\,h\sqrt{a+b\,x}} + \frac{(b\,c-a\,d)\,(b\,e-a\,f)\,(b\,g-a\,h)}{2\,b\,f\,h\,(a+b\,x)^{3/2}\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}} - \frac{(b\,d\,e\,h+f\,(b\,d\,g-b\,c\,h-a\,d\,h))\,\sqrt{a+b\,x}}{2\,b\,f\,h\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}$

Rule 1.1.1.4.2.2.2.1.1:

$$\int \frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx \longrightarrow$$

$$\frac{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{g+h\,x}}{h\sqrt{e+f\,x}} + \frac{(d\,e-c\,f)\,(b\,f\,g+b\,e\,h-2\,a\,f\,h)}{2\,f^2\,h}\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx +$$
$$\frac{(a\,d\,f\,h-b\,(d\,f\,g+d\,e\,h-c\,f\,h))}{2\,f^2\,h}\int \frac{\sqrt{e+f\,x}}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}\,\sqrt{g+h\,x}}\,dx - \frac{(d\,e-c\,f)\,(f\,g-e\,h)}{2\,f\,h}\int \frac{\sqrt{a+b\,x}}{\sqrt{c+d\,x}\,(e+f\,x)^{3/2}\,\sqrt{g+h\,x}}\,dx$$

Program code:

```
Int[Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]/(Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[g+h*x]/(h*Sqrt[e+f*x]) +
  (d*e-c*f)*(b*f*g+b*e*h-2*a*f*h)/(2*f^2*h)*Int[1/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  (a*d*f*h-b*(d*f*g+d*e*h-c*f*h))/(2*f^2*h)*Int[Sqrt[e+f*x]/(Sqrt[a+b*x]*Sqrt[c+d*x]*Sqrt[g+h*x]),x] -
  (d*e-c*f)*(f*g-e*h)/(2*f*h)*Int[Sqrt[a+b*x]/(Sqrt[c+d*x]*(e+f*x)^(3/2)*Sqrt[g+h*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

2: $\int \frac{(a+b x)^m \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx$ when $2 m \in \mathbb{Z} \land m > 1$

Rule 1.1.1.4.2.2.2.1.2: If $2 m \in \mathbb{Z} \land m > 1$, then

$$\int \frac{(a+b x)^m \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx \to$$

$$\frac{2 b (a+b x)^{m-1} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{f h (2 m+1)} - \frac{1}{f h (2 m+1)} \int \frac{(a+b x)^{m-2}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \cdot$$
$$\Big(a b (d e g+c (f g+e h)) + 2 b^2 c e g (m-1) - a^2 c f h (2 m+1) +$$
$$\left(b^2 (2 m-1) (d e g+c (f g+e h)) - a^2 d f h (2 m+1) + 2 a b (d f g+d e h-2 c f h m)\right) x -$$
$$b (a d f h (4 m-1) + b (c f h-2 d (f g+e h) m)) x^2\Big) \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*Sqrt[c_.+d_.*x_]/(Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  2*b*(a+b*x)^(m-1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/(f*h*(2*m+1)) -
  1/(f*h*(2*m+1))*Int[((a+b*x)^(m-2)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[a*b*(d*e*g+c*(f*g+e*h))+2*b^2*c*e*g*(m-1)-a^2*c*f*h*(2*m+1) +
    (b^2*(2*m-1)*(d*e*g+c*(f*g+e*h))-a^2*d*f*h*(2*m+1)+2*a*b*(d*f*g+d*e*h-2*c*f*h*m))*x -
    b*(a*d*f*h*(4*m-1)+b*(c*f*h-2*d*(f*g+e*h)*m))*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && GtQ[m,1]
```

2. $\int \frac{(a+b x)^m \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx$ when $2 m \in \mathbb{Z} \land m < 0$

1: $\int \frac{\sqrt{c+d x}}{(a+b x) \sqrt{e+f x} \sqrt{g+h x}} \, dx$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{c+d x}}{a+b x} = \frac{d}{b \sqrt{c+d x}} + \frac{b c-a d}{b (a+b x) \sqrt{c+d x}}$

Rule 1.1.1.4.2.2.2.2.1:

$$\int \frac{\sqrt{c+d x}}{(a+b x) \sqrt{e+f x} \sqrt{g+h x}} \, dx \to \frac{d}{b} \int \frac{1}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx + \frac{b c-a d}{b} \int \frac{1}{(a+b x) \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \, dx$$

Program code:

```
Int[Sqrt[c_.+d_.*x_]/((a_.+b_.*x_)*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  d/b*Int[1/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] +
  (b*c-a*d)/b*Int[1/((a+b*x)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),x] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

x: $\int \frac{\sqrt{c+d\,x}}{(a+b\,x)^{3/2}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{\sqrt{c+d\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}{\sqrt{e+f\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}} = 0$

- Basis: $\frac{\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}}{(a+b\,x)^{3/2}\,\sqrt{g+h\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}} = -\frac{2}{b\,g-a\,h}\,\text{Subst}\left[\frac{\sqrt{1+\frac{(b\,c-a\,d)\,x^2}{d\,g-c\,h}}}{\sqrt{1+\frac{(b\,e-a\,f)\,x^2}{f\,g-e\,h}}},\,x,\,\frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}\right]\partial_x \frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}$

Rule 1.1.1.4.2.2.2.2.2:

$$\int \frac{\sqrt{c+d\,x}}{(a+b\,x)^{3/2}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx \to \frac{\sqrt{c+d\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}{\sqrt{e+f\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}}\int \frac{\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}}{(a+b\,x)^{3/2}\,\sqrt{g+h\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}\,dx$$

$$\to -\frac{2\,\sqrt{c+d\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(e+f\,x)}{(f\,g-e\,h)\,(a+b\,x)}}}{(b\,g-a\,h)\,\sqrt{e+f\,x}\,\sqrt{\frac{(b\,g-a\,h)\,(c+d\,x)}{(d\,g-c\,h)\,(a+b\,x)}}}\,\text{Subst}\left[\int \frac{\sqrt{1+\frac{(b\,c-a\,d)\,x^2}{d\,g-c\,h}}}{\sqrt{1+\frac{(b\,e-a\,f)\,x^2}{f\,g-e\,h}}}\,dx,\,x,\,\frac{\sqrt{g+h\,x}}{\sqrt{a+b\,x}}\right]$$

Program code:

```
(* Int[Sqrt[c_.+d_.*x_]/((a_.+b_.*x_)^(3/2)*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -2*Sqrt[c+d*x]*Sqrt[(b*g-a*h)*(e+f*x)/((f*g-e*h)*(a+b*x))]/
    ((b*g-a*h)*Sqrt[e+f*x]*Sqrt[(b*g-a*h)*(c+d*x)/((d*g-c*h)*(a+b*x))])*
    Subst[Int[Sqrt[1+(b*c-a*d)*x^2/(d*g-c*h)]/Sqrt[1+(b*e-a*f)*x^2/(f*g-e*h)],x],x,Sqrt[g+h*x]/Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x] *)
```

2: $\int \frac{\sqrt{c+d x}}{(a+b x)^{3/2} \sqrt{e+f x} \sqrt{g+h x}} \, dx$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{\sqrt{c+d x} \sqrt{-\frac{(b e-a f) (g+h x)}{(f g-e h) (a+b x)}}}{\sqrt{g+h x} \sqrt{\frac{(b e-a f) (c+d x)}{(d e-c f) (a+b x)}}} = 0$

- **Basis:** $\frac{\sqrt{\frac{(b e-a f) (c+d x)}{(d e-c f) (a+b x)}}}{(a+b x)^{3/2} \sqrt{e+f x} \sqrt{-\frac{(b e-a f) (g+h x)}{(f g-e h) (a+b x)}}} = -\frac{2}{b e-a f} \text{Subst}\left[\frac{\sqrt{1+\frac{(b c-a d) x^2}{d e-c f}}}{\sqrt{1-\frac{(b g-a h) x^2}{f g-e h}}}, x, \frac{\sqrt{e+f x}}{\sqrt{a+b x}}\right] \partial_x \frac{\sqrt{e+f x}}{\sqrt{a+b x}}$

Rule 1.1.1.4.2.2.2.2.2:

$$\int \frac{\sqrt{c+d x}}{(a+b x)^{3/2} \sqrt{e+f x} \sqrt{g+h x}} \, dx \to \frac{\sqrt{c+d x} \sqrt{-\frac{(b e-a f) (g+h x)}{(f g-e h) (a+b x)}}}{\sqrt{g+h x} \sqrt{\frac{(b e-a f) (c+d x)}{(d e-c f) (a+b x)}}} \int \frac{\sqrt{\frac{(b e-a f) (c+d x)}{(d e-c f) (a+b x)}}}{(a+b x)^{3/2} \sqrt{e+f x} \sqrt{-\frac{(b e-a f) (g+h x)}{(f g-e h) (a+b x)}}} \, dx$$

$$\to -\frac{2 \sqrt{c+d x} \sqrt{-\frac{(b e-a f) (g+h x)}{(f g-e h) (a+b x)}}}{(b e-a f) \sqrt{g+h x} \sqrt{\frac{(b e-a f) (c+d x)}{(d e-c f) (a+b x)}}} \text{Subst}\left[\int \frac{\sqrt{1+\frac{(b c-a d) x^2}{d e-c f}}}{\sqrt{1-\frac{(b g-a h) x^2}{f g-e h}}} \, dx, x, \frac{\sqrt{e+f x}}{\sqrt{a+b x}}\right]$$

Program code:

```
Int[Sqrt[c_.+d_.*x_]/((a_.+b_.*x_)^(3/2)*Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  -2*Sqrt[c+d*x]*Sqrt[-(b*e-a*f)*(g+h*x)/((f*g-e*h)*(a+b*x))]/
    ((b*e-a*f)*Sqrt[g+h*x]*Sqrt[(b*e-a*f)*(c+d*x)/((d*e-c*f)*(a+b*x))])*
    Subst[Int[Sqrt[1+(b*c-a*d)*x^2/(d*e-c*f)]/Sqrt[1-(b*g-a*h)*x^2/(f*g-e*h)],x],x,Sqrt[e+f*x]/Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d,e,f,g,h},x]
```

3: $\int \frac{(a+b x)^m \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx$ when $2 m \in \mathbb{Z} \land m \leq -2$

Rule 1.1.1.4.2.2.2.2.3: If $2 m \in \mathbb{Z} \land m \leq -2$, then

$$\int \frac{(a+b x)^m \sqrt{c+d x}}{\sqrt{e+f x} \sqrt{g+h x}} \, dx \longrightarrow$$

$$\frac{b (a+b x)^{m+1} \sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}}{(m+1) (b e-a f) (b g-a h)}+\frac{1}{2 (m+1) (b e-a f) (b g-a h)} \int \frac{(a+b x)^{m+1}}{\sqrt{c+d x} \sqrt{e+f x} \sqrt{g+h x}} \cdot$$
$$\left(2 a c f h (m+1)-b (d e g+c (2 m+3) (f g+e h))+2 (a d f h (m+1)-b (m+2) (d f g+d e h+c f h)) x-b d f h (2 m+5) x^2\right) dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*Sqrt[c_.+d_.*x_]/(Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  b*(a+b*x)^(m+1)*Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]/((m+1)*(b*e-a*f)*(b*g-a*h)) +
  1/(2*(m+1)*(b*e-a*f)*(b*g-a*h))*Int[((a+b*x)^(m+1)/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]))*
    Simp[2*a*c*f*h*(m+1)-b*(d*e*g+c*(2*m+3)*(f*g+e*h))+2*(a*d*f*h*(m+1)-b*(m+2)*(d*f*g+d*e*h+c*f*h))*x-b*d*f*h*(2*m+5)*x^2,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m},x] && IntegerQ[2*m] && LeQ[m,-2]
```

3: $\int \frac{(e+f x)^p (g+h x)^q}{(a+b x) (c+d x)} \, dx$ when $0<p<1$

Derivation: Algebraic expansion

Basis: $\frac{e+f x}{(a+b x) (c+d x)} == \frac{b e-a f}{(b c-a d) (a+b x)} - \frac{d e-c f}{(b c-a d) (c+d x)}$

Rule 1.1.1.4.3: If $0<p<1$, then

$$\int \frac{(e+f x)^p (g+h x)^q}{(a+b x) (c+d x)} \, dx \longrightarrow \frac{b e-a f}{b c-a d} \int \frac{(e+f x)^{p-1} (g+h x)^q}{a+b x} \, dx - \frac{d e-c f}{b c-a d} \int \frac{(e+f x)^{p-1} (g+h x)^q}{c+d x} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^p_*(g_.+h_.*x_)^q_/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  (b*e-a*f)/(b*c-a*d)*Int[(e+f*x)^(p-1)*(g+h*x)^q/(a+b*x),x] -
  (d*e-c*f)/(b*c-a*d)*Int[(e+f*x)^(p-1)*(g+h*x)^q/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,q},x] && LtQ[0,p,1]
```

4: $\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx$ when $m \in \mathbb{Z}\,\bigwedge\,n+\frac{1}{2} \in \mathbb{Z}$

Derivation: Algebraic expansion

- **Rule 1.1.1.4.4: If** $m \in \mathbb{Z}\,\bigwedge\,n+\frac{1}{2} \in \mathbb{Z}$**, then**

$$\int \frac{(a+b\,x)^m\,(c+d\,x)^n}{\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,dx \longrightarrow \int \frac{1}{\sqrt{c+d\,x}\,\sqrt{e+f\,x}\,\sqrt{g+h\,x}}\,\text{ExpandIntegrand}\left[(a+b\,x)^m\,(c+d\,x)^{n+\frac{1}{2}},\,x\right]dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_/(Sqrt[e_.+f_.*x_]*Sqrt[g_.+h_.*x_]),x_Symbol] :=
  Int[ExpandIntegrand[1/(Sqrt[c+d*x]*Sqrt[e+f*x]*Sqrt[g+h*x]),(a+b*x)^m*(c+d*x)^(n+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h},x] && IntegerQ[m] && IntegerQ[n+1/2]
```

5: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$ when $(p\mid q) \in \mathbb{Z}$

Derivation: Algebraic expansion

- **Rule 1.1.1.4.5: If** $(p\mid q) \in \mathbb{Z}$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q,\,x\right]dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_)^q_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && IntegersQ[p,q]
```

6: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$ when $q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** $g+h\,x == \frac{h\,(a+b\,x)}{b} + \frac{b\,g-a\,h}{b}$

Rule 1.1.1.4.6: If $q \in \mathbb{Z}^+$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx \longrightarrow$$

$$\frac{h}{b}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^{q-1}\,dx+\frac{b\,g-a\,h}{b}\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^{q-1}\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_)^q_,x_Symbol] :=
  h/b*Int[(a+b*x)^(m+1)*(c+d*x)^n*(e+f*x)^p*(g+h*x)^(q-1),x] +
  (b*g-a*h)/b*Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^(q-1),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && IGtQ[q,0] && (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]] && Not[SumSimplerQ[p,1]])
```

C: $\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$

Rule 1.1.1.4.C:

$$\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx\longrightarrow\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*(e_.+f_.*x_)^p_.*(g_.+h_.*x_)^q_.,x_Symbol] :=
  CannotIntegrate[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x]
```

S: $\int (a+b\,u)^m\,(c+d\,u)^n\,(e+f\,u)^p\,(g+h\,u)^q\,dx$ when $u==i+j\,x$

Derivation: Integration by substitution

Rule 1.1.1.4.S: If $u==i+j\,x$, then

$$\int (a+b\,u)^m\,(c+d\,u)^n\,(e+f\,u)^p\,(g+h\,u)^q\,dx\longrightarrow\frac{1}{j}\,\text{Subst}\Big[\int (a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\,dx,\,x,\,u\Big]$$

Program code:

```
Int[(a_.+b_.*u_)^m_.*(c_.+d_.*u_)^n_.*(e_.+f_.*u_)^p_.*(g_.+h_.*u_)^q_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q,x],x,u] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p,q},x] && LinearQ[u,x] && NeQ[u,x]
```

Rules for integrands of the form $\left((a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\right)^r$

1: $\int\left((a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\right)^r\,dx$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(i\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\right)^r}{(a+b\,x)^{m\,r}\,(c+d\,x)^{n\,r}\,(e+f\,x)^{p\,r}\,(g+h\,x)^{q\,r}} = 0$

Rule:

$$\int\left(i\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\right)^r\,dx \longrightarrow$$
$$\frac{\left(i\,(a+b\,x)^m\,(c+d\,x)^n\,(e+f\,x)^p\,(g+h\,x)^q\right)^r}{(a+b\,x)^{m\,r}\,(c+d\,x)^{n\,r}\,(e+f\,x)^{p\,r}\,(g+h\,x)^{q\,r}}\int(a+b\,x)^{m\,r}\,(c+d\,x)^{n\,r}\,(e+f\,x)^{p\,r}\,(g+h\,x)^{q\,r}\,dx$$

Program code:

```
Int[(i_.*(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_*(e_.+f_.*x_)^p_*(g_.+h_.*x_)^q_)^r_,x_Symbol] :=
  (i*(a+b*x)^m*(c+d*x)^n*(e+f*x)^p*(g+h*x)^q)^r/((a+b*x)^(m*r)*(c+d*x)^(n*r)*(e+f*x)^(p*r)*(g+h*x)^(q*r))*
    Int[(a+b*x)^(m*r)*(c+d*x)^(n*r)*(e+f*x)^(p*r)*(g+h*x)^(q*r),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,m,n,p,q,r},x]
```

Normalize linear products

1: $\int u^m\,dx$ when $u == a+b\,x$

Derivation: Algebraic normalization

Rule: If $u == a+b\,x$**, then**

$$\int u^m\,dx \longrightarrow \int(a+b\,x)^m\,dx$$

Program code:

```
Int[u_^m_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m,x] /;
FreeQ[m,x] && LinearQ[u,x] && Not[LinearMatchQ[u,x]]
```

2: $\int u^m v^n \, dx$ when $u == a + b x \wedge v == c + d x$

Derivation: Algebraic normalization

Rule: If $u == a + b x \wedge v == c + d x$, then

$$\int u^m v^n \, dx \longrightarrow \int (a + b x)^m (c + d x)^n \, dx$$

Program code:

```
Int[u_^m_.*v_^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n,x] /;
FreeQ[{m,n},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

3: $\int u^m v^n w^p \, dx$ when $u == a + b x \wedge v == c + d x \wedge w == e + f x$

Derivation: Algebraic normalization

Rule: If $u == a + b x \wedge v == c + d x \wedge w == e + f x$, then

$$\int u^m v^n w^p \, dx \longrightarrow \int (a + b x)^m (c + d x)^n (e + f x)^p \, dx$$

Program code:

```
Int[u_^m_.*v_^n_.*w_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n*ExpandToSum[w,x]^p,x] /;
FreeQ[{m,n,p},x] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

4: $\int u^m v^n w^p z^q \, dx$ when $u == a + b x \land v == c + d x \land w == e + f x \land z == g + h x$

- **Derivation: Algebraic normalization**

- **Rule: If $u == a + b x \land v == c + d x \land w == e + f x \land z == g + h x$, then**

$$\int u^m v^n w^p z^q \, dx \longrightarrow \int (a + b x)^m (c + d x)^n (e + f x)^p (g + h x)^q \, dx$$

- **Program code:**

```
Int[u_^m_.*v_^n_.*w_^p_.*z_^q_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^n*ExpandToSum[w,x]^p*ExpandToSum[z,x]^q,x] /;
FreeQ[{m,n,p,q},x] && LinearQ[{u,v,w,z},x] && Not[LinearMatchQ[{u,v,w,z},x]]
```